

62-8000
570815475
5.95

TRS-80™

COLOUR COMPUTER

FANTASTIC GAMES



Radio Shack®

FANTASTIC GAMES

Programs Adapted by John Miller and Scott McCann

Editor: Carol Batchelor, B.A., M.A.

Illustration: Tim O'Halloran,
Margaret Horvath

Design: Margaret Horvath,
Pat Strachan

Cover: Richard Livingston

NOTE


The publisher and authors have made every effort to assure that the computer programs are accurate and complete. However, this publication is prepared for general readership, and neither the publisher nor the authors have any knowledge about or ability to control any third party's use of the programs and programming information. There is no warranty or representation by either the publisher or the authors that the programs or programming information in this book will enable the reader or user to achieve any particular result.

Copyright© 1984 by Hayes Publishing Ltd. All rights reserved. No part of this book may be reproduced or transmitted in any form or by any means, electronic or mechanical, including photocopying, recording or by any information storage and retrieval system, without permission from the publisher.

©1984 Hayes Publishing Ltd.
Burlington, Ontario


ISBN 0-88625-074-9


TRS-80™ is a registered trademark of Radio Shack, Division, Tandy Corporation



CONTENTS

Introduction	4
SPEEDBOAT	6
LOGGER	10
HAZE MAZE	16
GETAWAY	20
SUB ATTACK	26
SNAILS' TRAILS	30
How A Game Is Made	34





INTRODUCTION

This book contains listings for games written in BASIC for the TRS-80 COLOR COMPUTER made by Radio Shack.

You probably know from your owner's manual that the question mark (?) is a short way of typing in PRINT. You can save some time by using it when you are typing in these programs.


Remember, accidents do happen;

you could type in many lines of a program only to have a power failure, or you could unknowingly RUN a program with mistakes that destroy the program. So SAVE copies of the game on tape or floppy disk as you enter it. Do so frequently. Once you have a final copy of the game you can record over the other versions on tape or delete the disk versions.

DEBUGGING

After typing in a program and SAVEing it, the TRS-80 may have trouble RUNning it. This happens when copying a program from paper to computer. Check that you entered all the program lines completely. A

common problem is the famous SYNTAX ERROR. If you get one, list the program line on the screen. Compare it to the book listing. You will probably see one of these problems:



1. Spelling error, or
2. Punctuation error (brackets, commas, colons or semi-colons missing), or
3. The number zero confused with the letter 'O' (or vice versa), or
4. The number 1 confused with the letter 'I' (or vice versa).

Fix it and try **RUNning** the program again. It may take several attempts to get all the errors out, but the work will

be worth it. Remember to save a final copy that has all the corrections made.


EXPERIMENT

If you have some programming knowledge do not be afraid to try changing the games. After each program description is a list of what the program variables are used for. And remember, there is nothing that you can enter into the TRS-80 by

program or through the keyboard that can do any permanent damage. If something goes wrong, you can either turn the TRS-80 off and on to begin again, or press the RESET button.

SPEED BOAT

By Steven Holiday



Row row row your boat gently down the stream...

Not this time! You have a speed boat instead of the row boat. It isn't too difficult to navigate at first; just use the left and right arrow keys to steer. Throughout the river are bonus numbers. By guiding the boat over these you will build up a higher score. After a score of 3000 points the river will narrow. At 4000 it narrows still more. As well as the points in the water there are logs. Hitting these will get you as far as taking the boat to shore will... nowhere.

Variables Used

1\$ (1-3)	- the logs
R\$, RA\$, RB\$	- the rivers
R3\$, R6\$	- the bonuses
T	- the river's position
M	- the boat's position
Y	- the tree's position
H	- checks boat's position



SPEEDBOAT

```
0 CLS:PRINT@75,"SPEEDBOAT":PRINT@224,"ARE YOU USING JOYSTICKS  
[Y/N]"  
1 INPUT J$:IF LEFT$(J$,1)="Y" THEN J=1  
5 PRINT"INSTRUCTIONS [Y/N]":INPUT I$:IF LEFT$(I$,1)="Y" THEN  
CLS:GOSUB 850  
10 GOSUB 700:PRINT@452,"PRESS <ENTER> TO BEGIN":IF INKEY$<>  
CHR$(13) THEN 10  
20 FOR S=1 TO 27:READ N,T:SOUND N,T:NEXT S  
30 T=RND(18):M=T+5:MO=1088+M:S=0:T2=0:C=1:MN=5  
40 FOR X=1 TO 16:PRINT TAB(T) R$:NEXT  
50 IF PEEK(343)=247 OR J=1 AND JOYSTK(0)<30 THEN M=M-1  
60 IF PEEK(344)=247 OR J=1 AND JOYSTK(0)>33 THEN M=M+1  
70 POKE 1054,64+C:POKE 1055,112+MN:H=PEEK(1120+M):POKE 1120  
+M,128:POKE MO,175  
80 IF H=115 OR H=118 THEN S=S+H-112:SOUND 200,1  
90 IF H=159 OR H=255 THEN 300  
100 S=S+3:MO=M+1088:IF P=1 THEN P=0:RETURN  
110 X=RND(9):IF S>3000 THEN R$=R$:T2=2:C=2:X=RND(6)  
120 IF S>4000 THEN R$=R$:T2=4:C=3  
130 IF T=0 AND X<4 THEN T=2  
131 IF T>18+T2 AND X>3 THEN T=T-2  
140 IF X<4 THEN T=T-1:PRINT TAB(T) R$  
150 IF X>3 AND X<7 THEN T=T+1:PRINT TAB(T) R$  
160 IF X=7 THEN PRINTTAB(T)L$(RND(3))  
170 IF X=8 THEN PRINT TAB(T)R3$  
180 IF X=9 THEN PRINT TAB(T)R6$  
190 P=1:GOSUB 50  
200 Y=RND(32):IF Y<T+1 OR Y>T+12-T2 THEN POKE 1471+Y,94
```




```


220 PRINT TAB(T) R$:GOTO 50
300 MN=MN-1:FOR Q=1 TO 2:FOR X=120 TO 1 STEP-10:SOUNDX,1:
NEXTX,Q
310 IF MN<>0 THEN T=RND(18):M=T+5:MO=1088+M:GOTO 40
320 CLS:PRINT@98,"YOUR SCORE IS...";S:IF S>HG THEN HG=S
340 PRINT@194,"HIGH SCORE IS...";HG:PRINT@290,"ANOTHER GAME
[Y/N]"
350 G$=INKEY$:IF G$="" THEN 350
360 IF G$="Y" THEN RESTORE:GOTO 10
370 IF G$="N" THEN END ELSE 350
700 B$=CHR$(159)+CHR$(159):C$=CHR$(175):D$=CHR$(255)
710 R$="":L$="":FOR X=1 TO 8:R$=R$+C$:NEXT:R$=B$+R$+B$
720 RA$=B$+C$+C$:RB$=C$+C$+B$
730 R6$=RA$+C$+C$+"6"+C$+RB$ :R3$=RA$+C$+"3"+C$+C$+RB$
740 RB$=RA$+RB$:RA$=RA$+RIGHT$(RB$,6)
750 FOR X=1 TO 6:L$=L$+D$:NEXT
760 L$(1)=B$+C$+L$+C$+B$
770 L$(2)=B$+C$+D$+C$+RIGHT$(L$(1),7)
780 L$(3)=LEFT$(L$(1),7)+C$+D$+C$+B$
790 RETURN
800 DATA 89,5,89,5,89,5,108,4,125,4,125,3,108,3,125,3,133,
3,147,3,176,3,176,3,176,3,147,3,147,3,147,3,125,3,125,
3,125,3,89,3,89,3,89,3,147,3,133,3,125,3,108,3,89,3
850 PRINT@64,"IT'S A GREAT DAY TO GO CRUISING IN YOUR NEW
SPEEDBOAT."
860 PRINT"BUT BEWARE, IT'S LOGGING SEASON!"
870 PRINT@192,"STEER YOUR BOAT WITH THE (L,R) ARROW KEYS."
880 PRINT@288,"THE NUMBERS ARE BONUSSES.":RETURN

```



LOGGER

By Brian Jaikens



Picture yourself as a lumberjack out in the great redwood forests. You are cutting one down when suddenly, from out of nowhere, an eagle swoops down and steals your axe. You chase the fiendish bird back to its tree. In order to get your axe back, you must climb up into the bird's tree house. You use the arrow keys for

movement. But climbing is not the only challenge; all the time you are on the way up, logs are being rolled down at you. To jump over these logs use the space bar.

Variables Used

F\$, T\$, U\$,

V\$, W\$

L\$

A(X)

H

Q

- the logger's feet
- a ladder rung
- the log's position
- the logger's position
- the number of dead loggers



LOGGER



```
1 CLS:PRINT@128,"INSTRUCTIONS [Y/N]":INPUT I$:CLS:IF LEFT$(I$,1)="Y" GOSUB 850
10 DIM A(98):A(0)=1501:C$=CHR$(128):E$=C$+C$+C$:L$=CHR$(158)+CHR$(156)+CHR$(157)
20 F$="i"+C$+"i":T$=C$+"j"+C$:U$="j"+C$+"j":V$=C$+"l"+C$:W$="l"+C$+"l"
30 DATA 2,50,3,50,34,62,35,62,36,123,38,29,64,60,65,62,66,62,67,62,68,62,69,61,70,23,97,60,98,62,99,62,100,61,130,58,131,58
40 FOR X=1 TO 28:A(X)=1155+X:A(31+X)=1342-X:A(X+64)=1473+X:NEXT
50 FOR X=1 TO 5:A(X+26)=1182+32*X:A(X+59)=1313+32*X:NEXT X
60 RESTORE:H=475:OH=477:CLS 0:FOR X=1 TO 19:READ A,B:POKE 1024+A,129+B:NEXT
70 FOR X=1 TO 32:FOR Y=0 TO 320 STEP 160:POKE X+Y+1183,175:NEXT Y,X:FOR X=1 TO 3:POKE X+1212,128:POKE X+1343,128:NEXT X
80 FOR X=1 TO 3:PRINT@153+X*32,L$;:PRINT@292+X*32,L$;:NEXT X
90 SOUND 200,10:GOTO 600
100 IF H=448 OR H=134 THEN RETURN ELSE H=H-1
115 IF H>449 AND H<453 OR H>310 AND H<314 THEN PRINT@388,L$;:PRINT@249,L$; ELSE PRINT@H-62,C$;
120 IF INT(H/2)=H/2 THEN PRINT@H,U$; ELSE PRINT@H,T$;
130 PRINT@H-29,C$;:PRINT@H+3,C$;:RETURN
150 IF H=477 OR H=317 THEN RETURN ELSE H=H+1
165 IF H>312 AND H<316 OR H>451 AND H<455 THEN PRINT@388,L$;:PRINT@249,L$; ELSE PRINT@H-64,C$;
```




```

170 IF INT(H/2)=H/2 THEN PRINT@H,W$; ELSE PRINT@H,V$;
180 PRINT@H-33,C$;:PRINT@H-1,C$;:RETURN
200 IF PEEK(960+H)<>158 AND PEEK(1056+H)<>158 OR H=292 OR
    H=153 THEN RETURN
210 IF H=452 OR H=420 OR H=313 OR H=281 THEN PRINT@H,E$; ELSE
    PRINT@H,L$;
220 H=H-32:IF H=420 OR H=292 OR H=281 OR H=153 THEN PRINT@H,F$;
    ELSE PRINT@H,"i";CHR$(156);"i";
230 RETURN
250 IF PEEK(1056+H)<>158 AND PEEK(960+H)<>158 OR H=452 OR
    H=313 THEN RETURN
260 IF H=388 OR H=420 OR H=281 OR H=249 THEN PRINT@H-64,L$;
    :PRINT@H-32,L$; ELSE PRINT@H-63,C$;:PRINT@H-32,E$;
270 IF H=356 OR H=217 THEN PRINT@H-32,L$;
280 H=H+32:PRINT@H,"i";CHR$(PEEK(H+1025));"i";:RETURN
300 IF ABS(A(X)-1025-H)>3 OR H<455 AND H>449 OR H<316 AND
    H>310 THEN RETURN
310 PRINT@H,E$;
320 POKE 929+H,15:POKE 960+H,47:POKE 961+H,15:POKE 962+H,28:
    PRINT@H-32,F$;
330 FOR Y=1 TO 3:POKE A(X+Y-1),128:POKE A(X+Y),0
340 FOR J=1 TO 35:NEXT J,Y:S=S+80:IF PEEK(1094)=47 THEN 800
350 PRINT@H-96,E$;:PRINT@H-64,E$;:PRINT@H,F$;
360 OH=H+1:X=X+Y-1:SOUND 128,2:RETURN
600 X=1+N
610 PRINT@20,S;:POKE A(X-1-N),128:
    IF PEEK(A(X-N))<>128 OR H=157 OR
        H=288 THEN700

```





```

620 POKE A(X),0:IF X=2 THEN POKE 1155,185:POKE 1156,178
630 IF X=4 THEN POKE 1155,187:POKE 1156,128
640 FOR Y=1 TO 5:IF PEEK(340+Y)=247 THEN A=Y:Y=5
650 NEXT Y:ON A GOSUB 200,250,100,150,300:IF H=OH THEN PRINT
    @H,F$;:GOTO 680
670 POKE 961+H,15:POKE 992+H,47:POKE 993+H,15:POKE 994+H,28
680 S=S+OH-H:OH=H:A=0
690 IF X>60 AND H<320 OR X>27 AND H<160 OR X>=92 THEN POKE
    A(X),128:GOTO 600
695 X=X+1+N:GOTO 610
700 FOR K=H TO 447 STEP 32:GOSUB 260
710 POKE 961+H,15:POKE 992+H,47:POKE 993+H,15:POKE 994+H,28:
    SOUND 250-K/2,3
720 NEXT:PRINT@H-63,C$;:PRINT@H,"ioi";:FOR K=1 TO 1500:NEXT
730 IF Q<>3 THEN Q=Q+1:GOTO 60
740 CLS:PRINT@98,"SCORE...";S:IF S>HI THEN HI=S
750 PRINT@226,"HIGH...";HI:PRINT@416,"DO YOU WISH TO PLAY
    AGAIN [Y/N]?"
760 G$=INKEY$:IF G$="" THEN 760 ELSE IF G$="Y" THEN S=0:Q=0:
    N=0:GOTO 60

```




```

770 IF G$="N" THEN END ELSE GOTO 760
800 FOR P=1 TO 251 STEP 50:SOUND P,4:NEXT:S=S+920:N=1:GOTO 60
850 PRINT@64,"THE BIRD HAS STOLEN YOUR AXE AND YOU MUST GET
    IT BACK!"
860 PRINT@160,"GUIDE YOURSELF THROUGH THE TREE-HOUSE WITH
    THE ARROW KEYS."
870 PRINT"JUMP OVER THE LOGS WITH THE      SPACE BAR."
880 PRINT@320,"ONCE YOU HAVE REACHED THE BIRD  YOU MUST JUMP
    UP TO GET YOUR AXE."
890 PRINT@448,"PRESS A KEY TO CONTINUE
...":IF INKEY$="" THEN
    890 ELSE RETURN

```



Have you ever felt like a mouse in a maze? Well, with this game you might as well be one. But you won't be relying on your nose to find the cheese at the end because there is none. All you have to do is get out. You can see the maze exit at all times, and while it is being built you

have a chance to plan a way out. There is only one way out. Once the maze is complete, your 'mouse' appears in the top left-hand corner. Using the four arrow keys, you race against time to get to the lower right-hand corner of the screen. Hope you make it!

Variables Used

AS
MZ
R
X, Y
XX, YY

- input values
- the maze array
- random direction of maze setup
- positions in the maze
- screen positions

HAZE MAZE

By Mats Oehrman

adapted by Scott McCann



HAZE MAZE



```

10 REM "THIS PROGRAM REQUIRES
    16K EXTENDED COLOR BASIC"
40 DIM MZ(22,21):CLS 0
55 INPUT "INSTRUCTIONS [Y/N]";A$:IF A$="Y" THEN GOSUB5000
57 CLS 0
60 FORI=1TO20:MZ(0,I)=10:MZ(22,I)=10:NEXT I
70 FORI=1TO21:MZ(I,0)=5:MZ(I,21)=5:NEXT I
80 MZ(0,0)=15:MZ(0,21)=15:MZ(22,0)=5:MZ(22,21)=15
95 X=10:Y=10:GOSUB700
99 PMODE4,1:SCREEN1,1:PCLS:LINE(7,7)-(177,169),PSET,BF
100 A=0
110 IF MZ(X-1,Y)=0 THEN V(A)=1:A=A+1
120 IF MZ(X,Y-1)=0 THEN V(A)=4:A=A+1
130 IF MZ(X,Y+1)=0 THEN V(A)=2:A=A+1
140 IF MZ(X+1,Y)=0 THEN V(A)=3:A=A+1
150 IFA=0ANDX=10ANDY=10THEN1000
160 IF A=0 THEN R=INT(MZ(X,Y)/256):GOTO230
170 B=RND(A+1)-1:IF B>=A THEN 170
180 R=V(B)
190 IF R=1 THEN MZ(X,Y)=MZ(X,Y)OR1:MZ(X-1,Y)=772
192 IF R=2 THEN MZ(X,Y)=MZ(X,Y)OR2:MZ(X,Y+1)=1032
194 IF R=3 THEN MZ(X,Y)=MZ(X,Y)OR4:MZ(X+1,Y)=257
198 IF R=4 THEN MZ(X,Y)=MZ(X,Y)OR8:MZ(X,Y-1)=514
230 IF R=1THEN X=X-1
235 IF R=2THEN Y=Y+1
238 IF R=3THEN X=X+1
240 IF R=4THEN Y=Y-1
250 GOSUB 10000
270 GOTO 100
410 LINE(XX,YY)-(XX+7,YY),PSET:RETURN
420 LINE(XX,YY)-(XX,YY+7),PSET:RETURN
430 LINE(XX,YY)-(XX+7,YY),PSET:LINE(XX,YY)-(XX,YY+7),
    PSET:RETURN
700 XX=X*8:YY=Y*8:RETURN
1000 X=1:Y=1:MZ(1,1)=MZ(1,1)OR8:
    GOSUB10000
1010 X=1:Y=0:MZ(1,0)=2:GOSUB10000
1015 X=1:Y=0:MZ(1,0)=2:GOSUB10000
1020 X=1:Y=0:MZ(21,20)=MZ(21,20)OR2:
    GOSUB10000
1025 X=21:Y=21:MZ(21,21)=8:GOSUB
    10000
    
```




```


1030 X=1:Y=0:GOSUB 700:CIRCLE (XX+5,YY+5),2
1035 SOUND50,10:TIMER=0
1040 A$=INKEY$
1050 IF A$=CHR$(8)AND (MZ (X,Y)AND1)<>0 THENX=X-1:
GOTO 1085
1060 IF A$=CHR$(94)AND (MZ (X,Y)AND8)<>0 THENY=Y-1:
GOTO1085
1070 IF A$=CHR$(10)AND (MZ (X,Y)AND2)<>0 THENY=Y+1:
GOTO1085
1080 IF A$=CHR$(9)AND (MZ (X,Y)AND4)<>0 THENX=X+1:
GOTO1085
1081 IF A$="1" THEN SCREEN1,0:GOSUB 1205:SCREEN1,1
1083 GOTO 1040
1085 CIRCLE (XX+5,YY+5),2,0
1090 GOSUB700
1095 CIRCLE (XX+5,YY+5),2
1110 IFX<>22ANDY<>21THEN1040
1200 PLAY"L4Tl0A#A#DDDCCAEGFGACCDCAFGAAGG"
1205 PRINT0,"YOUR TIME IS";INT (TIMER/60*100)/100
1206 IF A$="1" THEN B$=INKEY$;IF B$="2" THENRETURN
ELSE GOTO1205
1210 INPUT"PLAY AGAIN [Y/N]";A$:IF A$="Y" THEN RUN
ELSE CLS:END
5000 CLS:PRINT"      haze maze instructions"
5020 PRINT065,"IN THIS GAME YOU MUST MOVE
      THROUGH A MAZE IN THE SHORTEST AMOUNT
      OF TIME."
5030 PRINT@192,"YOU ARE IN THE TOP LEFT CORNER.
      WHEN YOU HEAR THE BEEP USE";
5040 PRINT" THE ARROW KEYS TO GUIDE A MOUSE TO
      THE LOWER RIGHT CORNER."
5070 PRINT@352,"PRESS '1' FOR YOUR CURRENT TIME,
      AND '2' TO RETURN TO THE MAZE."
5080 PRINT@480,"PRESS <ENTER>";A$=INKEY$:IF A$=
CHR$(13) THEN RETURN ELSE 5080
10000 GOSUB700:COLOR0,0:LINE (XX,YY)-(XX+7,YY+7),
PSET,BF:COLOR5,0
10001 IF (MZ (X,Y)AND1)<>0 AND (MZ (X,Y)AND8)<>0
THEN RETURN
10010 IF (MZ (X,Y)AND1)<>0 THEN 410
10020 IF (MZ (X,Y)AND8)<>0 THEN 420
10030 GOTO 430

```




GETAWAY

By Howard Evans



You are the driver for a bank robbery. Sitting impatiently at the wheel of the van you await your colleagues' return. They are inside the bank throwing money out to you. As they are in the second story in several locations, you drive to the spots where they are throwing the money. Move the van with the left and right arrow keys.

Watch out though! Here come the police. Quickly you try to leave. In the rush you drop the van keys on the floor. You fumble about in the dark to pick the correct key to restart the engine before the police arrive.



Variables Used

T\$	- the van
SC	- the score
Z	- the money bag's position
H	- the van's position
U	- the car's position
HF	- the big van's position
W\$, X\$	- the big van
G\$	- the letter
L\$	- your letter guess
DD	- the number of men caught



GETAWAY



```

1 CLS:PRINT@128,"INSTRUCTIONS [Y/N]":INPUT I$:IF LEFT$(I$,1)
  ="Y" GOSUB 8500
50 DATA 5,244,58,239,89,232,125,227,147,218,170,204,185
100 CLS:PRINT@88,"$$$":PRINT@120,"$$$":PRINT@151,"$$$$$":
  PRINT@182,"$ $ $ $"
110 PRINT@213,"$ $ $":PRINT@246,"$$$$$$$":PRINT@278,"$$$$$$$":
  PRINT@316,"$ $ $"
120 PRINT@342,"$ $ $":PRINT@375,"$$$$$":PRINT@408,"$$$":
  PRINT@440,"$$$"
135 FOR Y=1 TO 3:RESTORE:FOR X=1 TO 13:
  V$=V$+CHR$(191):SOUND X,2:NEXT X,Y
140 PRINT@98,"'THE GETAWAY'";:PRINT@290,
  "PRESS <ENTER>";:PRINT@324,"TO BEGIN";
150 I$=INKEY$:IF I$<>CHR$(13) THEN 150
160 W$=LEFT$(V$,5):X$=LEFT$(V$,7)
170 R$=CHR$(128):CLS 0
200 FOR X=1024 TO 1119:POKE X,36:NEXT X
250 T$=LEFT$(V$,3)+CHR$(187):E$=R$+R$+R$+R$
260 PRINT@46,"bank";:PRINT@448,T$;
290 IF SC<2500 THEN ST=32:V1=0:P=1
300 F=RND(32)
310 FOR V=0+V1 TO 416 STEP ST
320 Z=1055+F+V:OL=PEEK(Z):POKE Z,100:OH=H
360 IF PEEK(343)=247 THEN H=H-1*P
370 IF PEEK(344)=247 THEN H=H+1*P
380 IF H<0 THEN H=0
390 IF H>28 THEN H=28
392 POKE 1055+F+V,OL
400 IF H<>OH THEN PRINT@448+OH,E$;:
  PRINT@448+H,T$;
  
```





```
440 NEXT V
460 IF OL=191 OR OL=187 THEN SOUND 170,1:SC=SC+120 ELSE
    SOUND 1,1
470 ST=64:V1=-32:P=2:SC=SC-20
490 SC$=STR$(SC):IF SC>-1 THEN SC$=RIGHT$(SC$,LEN(SC$)-1)
500 PRINT@78,"      ";:PRINT@78,SC$;
520 IF RND(16)=5 GOSUB 670
540 GOTO 290
670 HF=6+INT(H/3+.4)
680 PRINT@260,"THE POLICE ARE COMING !!";
690 FOR S=1 TO 3:SOUND 230,5:FOR SD=1 TO 100:NEXT SD,S
    730 FOR D=1 TO 999:NEXT:CLS0:T=0
    750 U=1313+T:POKE U,175:POKE U+1,172:POKE U+2,172:
        POKE U+3,175
    760 FOR X=31 TO 37:POKE U+X,175:NEXT X
    770 POKE U+38,152:POKE U+64,172:POKE U+68,172
    810 IF T>0 THEN 1000
    820 Y=264+HF:PRINT@Y,W$;
    830 PRINT@Y+32,W$;CHR$(188);CHR$(191);
    840 PRINT@Y+64,X$;CHR$(152);
    850 PRINT@Y+97,CHR$(188);:PRINT@Y+101,CHR$(188);
    900 IF EE=-1 THEN RETURN
    930 G$=CHR$(RND(26)+64)
    940 PRINT@0:PRINT@32,"GUESS THE RIGHT LETTER
        TO ESCAPE":PRINT
```




```

1000 L$=INKEY$:POKE U-30,179
1005 IF L$="" OR L$<"A" OR L$>"Z" THEN POKE U-30,128:GOTO
    1000
1010 PRINT@34+ASC(L$),L$;:SOUND 220,4:IF L$=G$ THEN 2000
1040 IF PEEK(U+39)=191 THEN 3000
1050 POKE U,128:POKE U+31,128:POKE U-30,128
1060 POKE U+68,128:POKE U+64,128:T=T+1:GOTO 750
2000 PRINT@32,"YOU ESCAPED, THE LETTER WAS ";L$
2015 SOUND 200,4:SOUND 220,3:SOUND 240,4
2020 FOR X=1 TO 999:NEXT:EE=-1:FOR Q=HF TO 15
2060 FOR X=0 TO 2:PRINT@Y+X*32,R$;:NEXT X
2070 PRINT@Y+101,R$;:PRINT@Y+97,R$;
2090 HF=HF+1:GOSUB 820:NEXT Q:H=0
2120 EE=0:FOR D=1 TO 1000:NEXT D:GOTO 170
3000 PRINT@64,"    THE CORRECT LETTER WAS ";G$
3010 DD=DD+1:IF DD=4 THEN 7000

```




```


3030 PRINT@32," THEY HAVE CAUGHT";DD;"OF YOUR MEN"
3050 FOR BB=101 TO 1 STEP -10:SOUND BB,4:NEXT BB
3080 FOR X=1 TO 800:NEXT X:H=0:GOTO 170
7000 PRINT@32,"          THEY'VE NAILED YOU !":PRINT@96
7010 FOR X=1 TO 800:NEXT:SOUND 55,5:SOUND 55,5:SOUND 55,5
7020 SOUND 1,12:IF SC>HI THEN HI=SC
7030 PRINT@32,"          THE HIGH SCORE IS";HI
7040 PRINT:PRINT"          PLAY AGAIN [Y/N]?":PRINT
7060 A$=INKEY$:IF A$=""THEN 7060
7080 IF A$="Y" THEN DD=0:SC=0:H=0:GOTO 170
7090 IF A$="N" THEN CLS ELSE 7060
8500 PRINT@96,"YOU DRIVE THE GETAWAY VAN."
8555 PRINT"YOU MUST CATCH THE MONEY YOUR  PARTNERS THROW
      OUT OF THE BANK."
8600 PRINT@256,"STEER THE VAN WITH THE (L,R)      ARROW KEYS."
8700 PRINT@448,"PRESS A KEY TO CONTINUE...":IF INKEY$=""
      THEN 8700 ELSE RETURN

```



SUB ATTACK


By Stuart Hall



The south coast is in danger. The waters are plagued by pirates. They travel the waters in many different vessels. You are the Captain of the Guardian, a submarine, and you have been selected to patrol the pirate-infested waters. The best strategy, you decide, is to sit on the ocean floor and use torpedoes to cripple the pirate ship. Your submarine's movement is controlled with the left and right arrow keys. Torpedoes are fired with the space bar. You have only one minute to destroy as much of the pirate fleet as you can.

Variables Used

B\$(1-8)	- the pirate ships
T\$(1-8)	- the blown up ships
B\$(9)	- the submarine
T\$(9)	- erases ships
T\$	- the trees
H	- the sub's position
Y	- the ship's position
F	- the torpedo's position
S	- the score



An illustration of a submarine and a missile in the ocean. The submarine is at the top left, firing a missile that is shown in mid-air, trailing a large plume of white smoke. The background is a greenish-blue gradient representing the water. At the bottom, there are several yellow and black striped fish swimming in the blue water, with some green seaweed or coral visible on the left.

SUB ATTACK

```
1 CLS:PRINT@128,"INSTRUCTIONS [Y/N]":INPUT I$:CLS:IF LEFT$(I$,1)="Y" GOSUB 850
10 Z=128:DATA 3,4,3,5,11,6,11,1,3,0,3,9,-1,17,0,18,28,19,
    22,-1,1,9,1,5,-1
15 DATA 113,116,-1,4,6,15,14,-60,9,-48,10,15,12,8,1,-1,
    29,28,31,22,31,26,31,29
20 DATA 28,20,-1,13,4,-48,9,-44,9,8,12,-1,125,118,127,
    117,126,124,-1,35,0,35,0
25 DATA 43,0,35,0,34,0,-1,:FOR B=1 TO 9
30 READ C
35 IF C=-1 THEN B$(B)=CHR$(Z)+B$(B):NEXT B:GOTO 50
40 READ D:B$(B)=B$(B)+CHR$(Z+C):T$(B)=T$(B)+CHR$(Z+D):GOTO 30
50 B$(9)=B$(9)+CHR$(Z):T$(9)=T$(9)+CHR$(Z)
60 CLS 0:FOR B=0 TO 190:PRINT@B,CHR$(175);:NEXT B
62 FOR B=1 TO 7:T$=T$+"^":IF B=3 THEN PRINT@94,T$;
64 IF B=5 THEN PRINT@125,T$;:PRINT@187,T$;
66 NEXT B:PRINT@156,T$;:H=490:PRINT@H,B$(9);
80 PRINT@237,"PRESS";:PRINT@269,"A KEY";
90 IF INKEY$="" THEN 90
100 PRINT@237,T$(9);:PRINT@269,T$(9);
110 R=RND(4)*32+191:B=RND(4):Y=R
```



```


120 X=X+1:PRINT@Y,B$(B);:PRINT@Y+32,B$(B+4);
130 IF PEEK(343)=247 AND H>481 THEN H=H-1
140 IF PEEK(344)=247 AND H<504 THEN H=H+1
150 PRINT@H,B$(9);:IF PEEK(345)=247 AND L=0 GOSUB 700
170 IF L=1 GOSUB 710
180 Y=Y+1:IF X=550 THEN 250
200 IF PEEK(R+1088)<>128 THEN POKE R+1088,128:PRINT@Y,T$(9)
;:PRINT@Y+32,T$(9);:GOTO 110
210 GOTO 120
250 IF S>HI THEN HI=S
260 PRINT@224,"HIGH SCORE...";HI;
270 PRINT@256,"YOUR SCORE...";S;
280 PRINT@320,"DO YOU WISH TO PLAY AGAIN [Y/N]?";:A$=INKEY$:IF
A$="" THEN 280
290 IF A$="Y" THEN S=0:X=0:L=0:T$="":GOTO 60
300 IF A$="N" THEN END ELSE 280
700 F=H+1027:L=1
710 F=F-32:P=PEEK(F):POKE F,30:IF F<1248 THEN L=0:GOTO 750
740 IF P<>128 THEN PRINT@Y+1,T$(B);:PRINT@Y+33,T$(B+4);
:L=0:S=S+B*20:SOUND 9,1:SOUND 120,2:POKE R+1088,32
750 POKE F,Z:RETURN
850 PRINT@32,"YOUR MISSION: TO DESTROY AS MA
NY PIRATE SHIPS AS POSSIBLE IN
ONE MINUTE."
860 PRINT@192,"THERE ARE FOUR TYPES OF SHIPS WORTH 20,40,60 A
ND 80 POINTS."
870 PRINT@288,"USE THE [L,R] ARROW KEYS TO GUIDE YOUR SUBMA
RINE, AND THE SPACE BAR TO FIRE TORPEDOES."
880 PRINT@448,"PRESS A KEY TO
CONTINUE...":IF INKEY$="" THEN 88
0 ELSE RETURN

```



SNAILS' TRAILS


By Peter Lear



This is a game for two players using joysticks or the keyboard. Player One, using the left side of the keyboard or the left joystick, starts at the top. Player Two starts at the bottom. The object of the game is to build up the highest score without running into the other player's trail or your own.

Variables Used

TM	- the counter
P(1,2)	- the snails' positions
S(1,2)	- the scores
D(1,2)	- the changes in position
N(1,2)	- checks the squares the snails are on



SNAILS' TRAILS

```

100 GOSUB 3000:GOSUB 800
110 P(1)=1103:P(2)=1520:S(1)=0:S(2)=0:D(1)=32:D(2)=-32
120 FOR TM=1 TO 600 STEP 3
140 IF PEEK(340)=253 OR J=1 AND JOYSTK(0)<30 THEN D(2)=-1
150 IF PEEK(341)=223 OR J=1 AND JOYSTK(0)>33 THEN D(2)=1
160 IF PEEK(339)=254 OR J=1 AND JOYSTK(2)<30 THEN D(1)=-1
170 IF PEEK(344)=254 OR J=1 AND JOYSTK(2)>33 THEN D(1)=1
180 IF PEEK(345)=253 OR J=1 AND JOYSTK(1)<30 THEN D(2)=-32
190 IF PEEK(342)=223 OR J=1 AND JOYSTK(1)>33 THEN D(2)=32
200 IF PEEK(343)=254 OR J=1 AND JOYSTK(3)<30 THEN D(1)=-32
210 IF PEEK(338)=247 OR J=1 AND JOYSTK(3)>33 THEN D(1)=32
220 FOR X=1 TO 2:P(X)=P(X)+D(X):N(X)=PEEK(P(X))
250 IF N(X)<>159 AND N(X)<>175 AND N(X)<>255 AND N(X)<>207
    THEN 6000
260 IF N(X)=159 THEN S(X)=S(X)+1
270 IF N(X)=175 THEN S(X)=S(X)+2
280 IF N(X)=255 THEN S(X)=S(X)+3
290 IF N(X)=207 THEN S(X)=S(X)+5
300 POKE P(1),185:POKE P(2),233
310 SOUND (127+D(X)*3),1:NEXT X
330 PRINT@34,S(1);:PRINT@41,TM;:PRINT@57,S(2);
340 NEXT TM
800 FOR C=1 TO 20:W$=W$+CHR$(207):R$=R$+CHR$(255):B$=B$+CHR$(175):NEXT C
810 M$=LEFT$(B$,4):K$=LEFT$(R$,3):C$=M$+K$+K$+K$+K$+M$:D$=M$+K$+LEFT$(W$,6)+K$+M$
820 CLS 2:PRINT"      ONE      TIME      HIGH      TWO"
830 PRINT"      0      0      0":PRINT@49,HI;
840 FOR X=-1 TO 0
850 PRINT@166+X*32,B$;:PRINT@422+X*32,B$;:SOUND 150,3
860 PRINT@230+X*32,C$;:PRINT@358+X*32,C$;:SOUND 30,3
870 PRINT@294+X*32,D$;:SOUND 200,3:NEXT X


```




```

890 FOR C=1088 TO 1504 STEP 32:POKE C,128:POKE C+31,128:
  NEXT C
900 POKE 1103,185:SOUND 30,7:POKE 1520,233:SOUND 220,7
910 RETURN
3000 CLS:PRINT@13,"SNAILS"
3010 PRINT@66,"POINTS DENOMINATION..."
3020 PRINT@98,CHR$(159);"  YELLOW =1"
3030 PRINT@130,CHR$(175);"  BLUE   =2"
3040 PRINT@162,CHR$(255);"  ORANGE =3"
3050 PRINT@194,CHR$(207);"  BUFF   =5"
3060 PRINT@258,"THE WINNER GETS A 20 PT. BONUS"
3070 PRINT@322,"ARE YOU USING JOYSTICKS [Y/N]?"
3080 J$=INKEY$:IF J$=""THEN 3080 ELSE IFJ$="Y"THEN J=1:
  GOTO 3140
3100 PRINT@322,"PLAYER ONE:      PLAYER TWO:"
3110 PRINT"      E              O"
3120 PRINT"      A      F      J      ;"
3130 PRINT"      X              ,"
3140 PRINT@485,"*PRESS ENTER TO BEGIN*";
3150 E$=INKEY$:IF E$<>CHR$(13) THEN 3150
3160 RETURN
6000 FOR S=100 TO 1 STEP-4
6010 SOUND S,2:NEXT S
6020 SOUND 1,12:S(3-X)=S(3-X)+20
6040 PRINT@134,"PLAYER ONE WINS !!!!!";
6050 IF S(2)>S(1) THEN PRINT@141,"TWO";
6060 IF S(2)>HI THEN HI=S(2)
6070 IF S(1)>HI THEN HI=S(1)
6075 PRINT@34,S(1);:PRINT@49,HI:PRINT@57,S(2)
6080 PRINT@262,"ANOTHER GAME [Y/N] ?";
6090 G$=INKEY$:IF G$="" THEN 6090
7000 IF G$="Y" THEN TM=601:GOSUB 820:GOTO 110
7010 IF G$="N" THEN CLS ELSE 6090

```

HOW A GAME IS MADE

Every game starts in the same place... in someone's head. The idea is then put down on paper. All the features of that game are written down. Pictures of the different characters and backgrounds are drawn. Every rule and aspect of the game is included in this paper plan.

The next step is to put all this information into an order of events. On another piece of paper shapes are drawn and each event of the game is put in a box, circle or diamond. With each figure a brief note is made. The name of this series of shapes and notes is a flowchart.

Every event in the flowchart is a small program in itself. These small programs are commonly called

subroutines. Breaking all the events into subroutines makes the task of programming the game much easier. Tracing a flaw in any particular subroutine is easier than tracing a flaw in a game made not using subroutines.

Quite often subroutines can be used more than once. They can even be transferred from one game to another. There is no point in designing a new subroutine to examine which way a player has moved the joystick for every game using a joystick. By using some of the same subroutines from game to game, a programmer will save himself/herself a lot of time.

Here are some guidelines to follow when designing a game:

1. Write down your ideas.
2. Draw up a flowchart.
3. Use subroutines.
4. Use the same subroutines where possible.

Here is a description of how the game SUB ATTACK works:

Line Number(s)	Operation
1	Gives instructions option.
10 to 50	Creates boats and submarine.
60 to 66	Sets up sky and coastline.
80 to 100	Waits for player to start.
110	Chooses both boat type and position.
120 to 210	Main loop.
120	Displays boat.
130 to 170	Moves and displays submarine; checks for torpedo fire.
180 to 200	Erases and redraws boat, and monitors time.
210	End of loop.
250 to 300	Scores routine and "play again" option.
700 to 750	Torpedo firing routine.
850 to 880	Instructions' display routine.

RADIO SHACK DIVISION, TANDY ELECTRONICS LIMITED

**CANADA: BARRIE, ONTARIO L4M 4W5
U.S.A.: FORT WORTH, TEXAS 76102**

TANDY CORPORATION

AUSTRALIA

91 KURRAJONG AVENUE
MOUNT DRUITT, N.S.W. 2770

U.K.

BILSTON ROAD, WEDNESBURY
WEST MIDLANDS WS10 7JN

BELGIUM

PARC INDUSTRIEL DE NANINNE
5140 NANINNE

